Miquel's Theorem in Isabelle/HOL

Arthur Freitas Ramos David Barros Hulak
Ruy J. G. B. de Queiroz

August 23, 2026

Abstract

We formalize Miquel's theorem, also known as the pivot theorem, in Isabelle/HOL. Let ABC be a triangle in the Euclidean plane and let P, Q, R be points on the side lines BC, CA, AB respectively. Then the circumcircles of the three triangles AQR, BRP, CPQ pass through a common point M, the *Miquel point* of the configuration.

The development uses complex coordinates.

Contents

1 Introduction **1**

2 The cross-ratio criterion **2**

3 The pivot theorem **2**

4 Existence of the circumscribed circle **3**

5 Concurrency: constructing the Miquel point **3**

6 Formalization notes **4**

7 Existence of the circumscribed circle **4**

8 Miquel's pivot theorem **8**

9 Concurrency of the three circumcircles **11**

1 Introduction

Miquel's theorem is a classical incidence result about a triangle and three points chosen on its side lines [4, 2, 1]. Given a triangle ABC and points P, Q, R on the lines BC, CA, AB, the three circumcircles of AQR, BRP

and CPQ are concurrent. Their common point M is the Miquel point, and the theorem is sometimes called the pivot theorem because M is the centre of the spiral similarity that pivots the configuration.

Our formalization works in the complex plane. It reuses the shared circle, line, collinearity, and cross-ratio interface from the published Wallace–Simson entry [5], which in turn builds on the published *Complex Geometry* development [3]. Reducing the incidence statement to complex algebra then turns the theorem into a short computation.

2 The cross-ratio criterion

Four complex numbers a, b, c, d have cross ratio

$$(a, b; c, d) = \frac{(a - c)\,(b - d)}{(a - d)\,(b - c)}.$$

The criterion we use is that, for four suitably distinct points, being concyclic or collinear is equivalent to the cross ratio being real. We only need the concyclic case, in both directions.

For the forward direction, if a, b, c, d lie on a circle of centre o and radius $R > 0$, then each point satisfies $\overline{z - o} = R^2/(z - o)$. Substituting this into the conjugate of the cross ratio and simplifying, every factor R^2 and every factor $(z - o)$ cancels, leaving the cross ratio itself. Hence the cross ratio equals its own conjugate and is real.

For the converse, suppose c, p, q lie on the circle of centre o and radius R, and that the cross ratio of a fourth point m with them is real. Writing $M = m - o$ and using the same conjugate substitution for c, p, q, the reality of the cross ratio simplifies to

$$(K - P)\left(M\,\overline{M} - R^2\right) = 0,$$

where $P = p - o$ and $K = q - o$. Since $p \neq q$ we have $K \neq P$, so $M\,\overline{M} = R^2$, that is $|m - o| = R$ and m lies on the same circle.

3 The pivot theorem

Let M be a common point of the circumcircles of AQR and BRP. By the forward criterion the cross ratios $(m, a; q, r)$ and $(m, b; r, p)$ are real. A direct computation shows

$$(m, a; q, r)\,(m, b; r, p)\,(m, c; p, q) = \frac{(a - r)\,(b - p)\,(c - q)}{(a - q)\,(b - r)\,(c - p)},$$

the point m cancelling completely. The right-hand side factors as

$$\frac{a - r}{b - r} \cdot \frac{b - p}{c - p} \cdot \frac{c - q}{a - q},$$

and each of the three ratios is real because R lies on line AB, P on line BC, and Q on line CA. Therefore $(m, c; p, q)$ is real, and by the converse criterion M lies on the circumcircle of CPQ. The three circumcircles are thus concurrent at M.

4 Existence of the circumscribed circle

The converse criterion needs the circle through C, P, Q to exist. Rather than assuming it, we construct it whenever the three points are not collinear. Working relative to a, set $u = b - a$ and $v = c - a$ and

$$D = \overline{u}\, v - u\, \overline{v}.$$

A short conjugation calculation shows $D = 2\,\mathrm{Im}(\overline{u}\, v)\, i$, so D is purely imaginary and $D \neq 0$ is exactly the non-collinearity of a, b, c. The point

$$o = a + w, \qquad w = \frac{u\, v\, \overline{(u - v)}}{D},$$

then satisfies $|a - o| = |b - o| = |c - o| = |w| > 0$; indeed $\overline{w} = \overline{u}\, \overline{v}\, (v - u)/D$, and the equalities follow from the two identities $w\, \overline{u} + \overline{w}\, u = u\, \overline{u}$ and $w\, \overline{v} + \overline{w}\, v = v\, \overline{v}$. Thus any three non-collinear points are concyclic.

To apply this to C, P, Q we note that, with P on line BC and Q on line CA, the cross product $\mathrm{Im}(\overline{(P - C)}\, (Q - C))$ is a nonzero real multiple of $\mathrm{Im}(\overline{(B - C)}\, (A - C))$; the latter is nonzero because ABC is nondegenerate. Hence C, P, Q are non-collinear and their circumcircle exists. The same computation shows $P \neq Q$, which the converse criterion also requires.

5 Concurrency: constructing the Miquel point

The pivot theorem assumes a point M already on two of the circumcircles. The classical statement instead asserts that such a point *exists*: the three circumcircles are concurrent. We prove this by constructing M.

The circumcircles of AQR and BRP, with centres o_1 and o_2, both pass through R. Two distinct circles meet in at most two points, and those points are symmetric about the line $o_1 o_2$. We therefore take M to be the reflection of R in that line,

$$M = o_1 + \frac{(o_2 - o_1)\, \overline{(R - o_1)}}{\overline{(o_2 - o_1)}}.$$

Reflection in the line through o_1 and o_2 is an isometry that fixes both centres, so $|M - o_1| = |R - o_1|$ and $|M - o_2| = |R - o_2|$; as R lies on both circles, so does M. When R is not on the line $o_1 o_2$ — that is, the two circles are not tangent at R — the reflection is genuinely distinct from R, and the pivot theorem places M on the third circumcircle as well. Should M happen to coincide with P or Q, concurrency is immediate because C, P, Q are non-collinear and hence concyclic. Thus the three circumcircles share the common point M.

6 Formalization notes

The development provides both forms. The pivot theorem *miquel-pivot* takes a point on two circumcircles and places it on the third. The concurrency theorem *miquel-concurrent* assumes no common point: it exhibits the Miquel point as the reflection above and shows it lies on all three circles, and the corollary *miquel-point-exists* states the pure existence form. The only structural hypothesis is that A, B, C are not collinear; from this the circumcircle of CPQ is constructed rather than assumed, and $P \neq Q$ is derived. The concurrency form additionally assumes the two circumcircles are not tangent at R (equivalently, R is not on the line through their centres), which is exactly the condition for a genuine second intersection. The remaining hypotheses are the three side-line conditions and general-position distinctness of the points involved, which guarantees that the cross ratios are defined and nonzero.

theory *Miquel*
  imports *Simson.Simson-Complex-Geometry*
begin

7 Existence of the circumscribed circle

Any three non-collinear points lie on a common circle of positive radius. We exhibit the circumcentre in closed form. Writing $u = b - a$ and $v = c - a$, the centre is $a + w$ with $w = u * v * cnj\ (u - v)\ /\ D$ and $D = cnj\ u * v - u * cnj\ v$. Non-collinearity is exactly $D \neq 0$, and a short conjugation computation shows the centre is equidistant from the three points.

lemma *circumcircle-exists*:
  assumes *ncol*: ¬ *collinear a b c*
  shows *concyclic3 a b c*
proof −
  define *u* **where** *u = b − a*
  define *v* **where** *v = c − a*
  define *D* **where** *D = cnj u ∗ v − u ∗ cnj v*
  have *imz*: *Im (cnj u ∗ v) ≠ 0* **using** *ncol* **by** (*simp add*: *collinear-iff-cross u-def v-def*)
  have *Dval*: *D = complex-of-real (2 ∗ Im (cnj u ∗ v)) ∗* i
    unfolding *D-def* **by** (*metis complex-cnj-cnj complex-cnj-mult complex-diff-cnj*)
  define *s* **where** *s = Im (cnj u ∗ v)*
  have *Dvals*: *D = complex-of-real (2 ∗ s) ∗* i **unfolding** *s-def* **by** (*rule Dval*)

have *imzs*: $s \neq 0$ **unfolding** *s-def* **by** (*rule imz*)
have *Dne*: $D \neq 0$ **using** *imzs* **by** (*simp add*: *Dvals*)
have *cnjD*: *cnj* $D = - D$ **unfolding** *D-def*
by (*metis complex-cnj-cnj complex-cnj-diff complex-cnj-mult minus-diff-eq*)
define w **where** $w = u * v *$ *cnj* $(u - v) / D$
have *cnjw*: *cnj* $w =$ *cnj* $u *$ *cnj* $v * (v - u) / D$
proof −
have *cnj* $w =$ *cnj* $u *$ *cnj* $v * (u - v) /$ *cnj* D
unfolding *w-def* **by** (*simp add*: *complex-cnj-mult complex-cnj-divide complex-cnj-diff*)
also have ... $=$ *cnj* $u *$ *cnj* $v * (u - v) / (- D)$ **by** (*simp add*: *cnjD*)
also have ... $=$ *cnj* $u *$ *cnj* $v * (v - u) / D$
proof −
have *cnj* $u *$ *cnj* $v * (u - v) / (- D) = -$ (*cnj* $u *$ *cnj* $v * (u - v) / D$) **by** *simp*
also have ... $= -$ (*cnj* $u *$ *cnj* $v * (u - v)) / D$ **by** *simp*
also have ... $=$ *cnj* $u *$ *cnj* $v * (v - u) / D$ **by** (*simp add*: *right-diff-distrib*)
finally show *?thesis* **.**
qed
finally show *?thesis* **.**
qed
have *wD*: $w * D = u * v *$ *cnj* $(u - v)$ **unfolding** *w-def* **using** *Dne* **by** *simp*
have *cnjwD*: *cnj* $w * D =$ *cnj* $u *$ *cnj* $v * (v - u)$ **unfolding** *cnjw* **using** *Dne* **by** *simp*
have *e1*: $w *$ *cnj* $u +$ *cnj* $w * u = u *$ *cnj* u
proof −
have $(w *$ *cnj* $u +$ *cnj* $w * u) * D = (w * D) *$ *cnj* $u +$ (*cnj* $w * D) * u$
by (*simp add*: *algebra-simps*)
also have ... $= u * v *$ *cnj* $(u - v) *$ *cnj* $u +$ *cnj* $u *$ *cnj* $v * (v - u) * u$
by (*simp add*: *wD cnjwD*)
also have ... $= u *$ *cnj* $u * D$ **unfolding** *D-def complex-cnj-diff* **by** *algebra*
finally have $(w *$ *cnj* $u +$ *cnj* $w * u) * D = u *$ *cnj* $u * D$ **.**
thus *?thesis* **using** *Dne* **by** (*metis mult-right-cancel*)
qed
have *e2*: $w *$ *cnj* $v +$ *cnj* $w * v = v *$ *cnj* v
proof −
have $(w *$ *cnj* $v +$ *cnj* $w * v) * D = (w * D) *$ *cnj* $v +$ (*cnj* $w * D) * v$
by (*simp add*: *algebra-simps*)
also have ... $= u * v *$ *cnj* $(u - v) *$ *cnj* $v +$ *cnj* $u *$ *cnj* $v * (v - u) * v$
by (*simp add*: *wD cnjwD*)
also have ... $= v *$ *cnj* $v * D$ **unfolding** *D-def complex-cnj-diff* **by** *algebra*
finally have $(w *$ *cnj* $v +$ *cnj* $w * v) * D = v *$ *cnj* $v * D$ **.**
thus *?thesis* **using** *Dne* **by** (*metis mult-right-cancel*)
qed
have *d1*: $(u - w) *$ *cnj* $(u - w) = w *$ *cnj* w
proof −
have $(u - w) *$ *cnj* $(u - w) = (u - w) *$ (*cnj* $u -$ *cnj* w) **by** (*simp add*: *complex-cnj-diff*)
also have ... $= u *$ *cnj* $u - (w *$ *cnj* $u +$ *cnj* $w * u) + w *$ *cnj* w

```
      by (simp add: algebra-simps)
    also have ... = w * cnj w using e1 by simp
    finally show ?thesis .
  qed
  have d2: (v - w) * cnj (v - w) = w * cnj w
  proof -
     have (v - w) * cnj (v - w) = (v - w) * (cnj v - cnj w) by (simp add:
complex-cnj-diff)
    also have ... = v * cnj v - (w * cnj v + cnj w * v) + w * cnj w
      by (simp add: algebra-simps)
    also have ... = w * cnj w using e2 by simp
    finally show ?thesis .
  qed
  have R1: cmod (u - w) = cmod w
  proof -
    have complex-of-real ((cmod (u - w))^2) = complex-of-real ((cmod w)^2)
    proof -
      have complex-of-real ((cmod (u - w))^2) = (u - w) * cnj (u - w)
        by (rule complex-norm-square)
      also have ... = w * cnj w by (rule d1)
     also have ... = complex-of-real ((cmod w)^2) by (rule complex-norm-square[symmetric])
      finally show ?thesis .
    qed
    hence (cmod (u - w))^2 = (cmod w)^2 by (metis of-real-eq-iff)
    thus ?thesis by (metis norm-ge-zero power2-eq-imp-eq)
  qed
  have R2: cmod (v - w) = cmod w
  proof -
    have complex-of-real ((cmod (v - w))^2) = complex-of-real ((cmod w)^2)
    proof -
      have complex-of-real ((cmod (v - w))^2) = (v - w) * cnj (v - w)
        by (rule complex-norm-square)
      also have ... = w * cnj w by (rule d2)
     also have ... = complex-of-real ((cmod w)^2) by (rule complex-norm-square[symmetric])
      finally show ?thesis .
    qed
    hence (cmod (v - w))^2 = (cmod w)^2 by (metis of-real-eq-iff)
    thus ?thesis by (metis norm-ge-zero power2-eq-imp-eq)
  qed
  have ab: a ≠ b
  proof
    assume a = b
    hence collinear a b c by (simp add: collinear-iff-cross)
    with ncol show False by simp
  qed
  have ac: a ≠ c
  proof
    assume a = c
    hence collinear a b c by (simp add: collinear-iff-cross)
```

with *ncol* **show** *False* **by** *simp*
qed
have *bc*: $b \neq c$
proof
assume *bc′*: $b = c$
have $Im\ (cnj\ (b - a) * (c - a)) = 0$
using *bc′* **by** (*metis complex-In-mult-cnj-zero mult.commute*)
hence *collinear a b c* **by** (*simp add*: *collinear-iff-cross*)
with *ncol* **show** *False* **by** *simp*
qed
have *u0*: $u \neq 0$ **using** *ab* **by** (*simp add*: *u-def*)
have *v0*: $v \neq 0$ **using** *ac* **by** (*simp add*: *v-def*)
have *uv*: $u \neq v$ **using** *bc* **by** (*simp add*: *u-def v-def*)
have *w0*: $w \neq 0$
proof
assume $w = 0$
hence $u * v * cnj\ (u - v) = 0$ **using** *wD* **by** *simp*
thus *False* **using** *u0 v0 uv* **by** (*simp add*: *complex-cnj-zero*)
qed
have *onA*: *oncircle* $(a + w)$ (*cmod w*) *a* **by** (*simp add*: *oncircle-def*)
have *onB*: *oncircle* $(a + w)$ (*cmod w*) *b*
proof −
have *cmod* $(b - (a + w))$ = *cmod* $(u - w)$ **by** (*simp add*: *u-def algebra-simps*)
also have … = *cmod w* **by** (*rule R1*)
finally show *?thesis* **by** (*simp add*: *oncircle-def*)
qed
have *onC*: *oncircle* $(a + w)$ (*cmod w*) *c*
proof −
have *cmod* $(c - (a + w))$ = *cmod* $(v - w)$ **by** (*simp add*: *v-def algebra-simps*)
also have … = *cmod w* **by** (*rule R2*)
finally show *?thesis* **by** (*simp add*: *oncircle-def*)
qed
have *Rpos*: $0 <$ *cmod w* **using** *w0* **by** *simp*
show *concyclic3 a b c*
unfolding *concyclic3-def*
using *onA onB onC Rpos* **by** *blast*
qed

Antisymmetry of the planar cross product under a change of base point.

lemma *cross-base-change*:
fixes *a b c* :: *complex*
shows $Im\ (cnj\ (b - c) * (a - c)) = -\ Im\ (cnj\ (b - a) * (c - a))$
by (*simp add*: *algebra-simps*)

If the triangle abc is nondegenerate and p, q lie on the side lines bc and ca without coinciding with c, then c, p, q are themselves non-collinear. Hence their circumscribed circle exists.

lemma *cpq-ncol*:
assumes *ncol*: ¬ *collinear a b c*

and *lp*: *on-line b c p*
and *lq*: *on-line c a q*
and *pc*: *p ≠ c*
and *qc*: *q ≠ c*
shows ¬ *collinear c p q*
proof −
from *lp* **obtain** *t* **where** *t*: *p* = *b* + *of-real t* ∗ (*c* − *b*) **using** *on-line-def* **by** *auto*
from *lq* **obtain** *s* **where** *s*: *q* = *c* + *of-real s* ∗ (*a* − *c*) **using** *on-line-def* **by** *auto*
have *pmc*: *p* − *c* = *of-real* (*1* − *t*) ∗ (*b* − *c*)
by (*simp add*: *t of-real-diff algebra-simps*)
have *qmc*: *q* − *c* = *of-real s* ∗ (*a* − *c*)
by (*simp add*: *s*)
have *imkey*: *Im* (*cnj* (*p* − *c*) ∗ (*q* − *c*)) = (*1* − *t*) ∗ *s* ∗ *Im* (*cnj* (*b* − *c*) ∗ (*a* − *c*))
by (*simp add*: *pmc qmc algebra-simps*)
have *t1*: *1* − *t* ≠ *0*
proof
assume *1* − *t* = *0*
hence *p* − *c* = *0* **by** (*simp add*: *pmc*)
thus *False* **using** *pc* **by** *simp*
qed
have *s0*: *s* ≠ *0*
proof
assume *s* = *0*
hence *q* − *c* = *0* **by** (*simp add*: *qmc*)
thus *False* **using** *qc* **by** *simp*
qed
have *imbc*: *Im* (*cnj* (*b* − *c*) ∗ (*a* − *c*)) ≠ *0*
proof −
have *ne*: *Im* (*cnj* (*b* − *a*) ∗ (*c* − *a*)) ≠ *0* **using** *ncol* **by** (*simp add*: *collinear-iff-cross*)
have *Im* (*cnj* (*b* − *c*) ∗ (*a* − *c*)) = − *Im* (*cnj* (*b* − *a*) ∗ (*c* − *a*)) **by** (*rule cross-base-change*)
with *ne* **show** *?thesis* **by** *simp*
qed
have *prod0*: (*1* − *t*) ∗ *s* ∗ *Im* (*cnj* (*b* − *c*) ∗ (*a* − *c*)) ≠ *0*
using *t1 s0 imbc* **by** *simp*
from *prod0 imkey* **have** *Im* (*cnj* (*p* − *c*) ∗ (*q* − *c*)) ≠ *0* **by** *simp*
thus ¬ *collinear c p q* **by** (*simp add*: *collinear-iff-cross*)
qed

8 Miquel's pivot theorem

The three cross ratios attached to the Miquel configuration multiply to a product of three ratios coming from the side lines.

lemma *miquel-complex-cross-ratio-product*:
assumes *dmr*: *m ≠ r* **and** *daq*: *a ≠ q* **and** *dmp*: *m ≠ p*

```
    and dbr: b ≠ r and dmq: m ≠ q and dcp: c ≠ p
  shows complex-cross-ratio m a q r * complex-cross-ratio m b r p * complex-cross-ratio
m c p q
       = (a − r) * (b − p) * (c − q) / ((a − q) * (b − r) * (c − p))
proof −
  have n1: m − r ≠ 0 using dmr by simp
  have n3: m − p ≠ 0 using dmp by simp
  have n5: m − q ≠ 0 using dmq by simp
  have complex-cross-ratio m a q r * complex-cross-ratio m b r p * complex-cross-ratio
m c p q
      = (((m−q)*(a−r)) * ((m−r)*(b−p)) * ((m−p)*(c−q)))
        / (((m−r)*(a−q)) * ((m−p)*(b−r)) * ((m−q)*(c−p)))
    unfolding complex-cross-ratio-def by (simp add: mult.commute mult.left-commute)
  also have ... = (((m−p)*(m−q)*(m−r)) * ((a−r)*(b−p)*(c−q)))
        / (((m−p)*(m−q)*(m−r)) * ((a−q)*(b−r)*(c−p)))
    by (simp add: mult.commute mult.left-commute)
  also have ... = (a − r) * (b − p) * (c − q) / ((a − q) * (b − r) * (c − p))
    using n1 n3 n5 by simp
  finally show ?thesis .
qed
```

The main theorem. With the side-line conditions on p, q, r, and general-position distinctness, any point m lying on the circumcircles of aqr and brp also lies on the circumcircle of cpq. Thus the three circumcircles are concurrent at the Miquel point m.

```
theorem miquel-pivot:
  assumes lr: on-line a b r and lp: on-line b c p and lq: on-line c a q
    and h1: concyclic a q r m and h2: concyclic b r p m
    and ncol: ¬ collinear a b c
    and dmp: m ≠ p and dmq: m ≠ q and dmr: m ≠ r
    and daq: a ≠ q and dar: a ≠ r and dbr: b ≠ r and dbp: b ≠ p
    and dcp: c ≠ p and dcq: c ≠ q
  shows concyclic c p q m
proof −
  — unpack the three circles
  from h1 obtain o1 R1 where c1: 0 < R1
      oncircle o1 R1 a oncircle o1 R1 q oncircle o1 R1 r oncircle o1 R1 m
    by (auto simp: concyclic-def)
  from h2 obtain o2 R2 where c2: 0 < R2
      oncircle o2 R2 b oncircle o2 R2 r oncircle o2 R2 p oncircle o2 R2 m
    by (auto simp: concyclic-def)
  — the third circle exists because the triangle is nondegenerate
  have ncolcpq: ¬ collinear c p q
    using cpq-ncol[OF ncol lp lq] dcp dcq by auto
  have dpq: p ≠ q
  proof
    assume p = q
    hence Im (cnj (p − c) * (q − c)) = 0
      by (metis complex-In-mult-cnj-zero mult.commute)
```

hence *collinear c p q* **by** (*simp add*: *collinear-iff-cross*)
with *ncolcpq* **show** *False* **by** *simp*
qed
from *circumcircle-exists*[*OF ncolcpq*] **obtain** *o3 R3* **where** *c3*: *0* < *R3*
oncircle o3 R3 c oncircle o3 R3 p oncircle o3 R3 q
by (*auto simp*: *concyclic3-def*)
— the two known circles make two cross ratios real
have *cr1*: *complex-cross-ratio m a q r* ∈ ℝ
using *concyclic-complex-cross-ratio-real*[*OF c1*(*5*) *c1*(*2*) *c1*(*3*) *c1*(*4*) *c1*(*1*)] *dmr daq* **by** *simp*
have *cr2*: *complex-cross-ratio m b r p* ∈ ℝ
using *concyclic-complex-cross-ratio-real*[*OF c2*(*5*) *c2*(*2*) *c2*(*3*) *c2*(*4*) *c2*(*1*)] *dmp dbr* **by** *simp*
— the product of all three cross ratios is a product of real side ratios
have *side1*: (*a* − *r*) / (*b* − *r*) ∈ ℝ
using *on-line-ratio-real*[*OF lr*] *dbr* **by** *simp*
have *side2*: (*b* − *p*) / (*c* − *p*) ∈ ℝ
using *on-line-ratio-real*[*OF lp*] *dcp* **by** *simp*
have *side3*: (*c* − *q*) / (*a* − *q*) ∈ ℝ
using *on-line-ratio-real*[*OF lq*] *daq* **by** *simp*
have *Phi*: (*a* − *r*) ∗ (*b* − *p*) ∗ (*c* − *q*) / ((*a* − *q*) ∗ (*b* − *r*) ∗ (*c* − *p*)) ∈ ℝ
proof −
have (*a* − *r*) ∗ (*b* − *p*) ∗ (*c* − *q*) / ((*a* − *q*) ∗ (*b* − *r*) ∗ (*c* − *p*))
= ((*a* − *r*) / (*b* − *r*)) ∗ ((*b* − *p*) / (*c* − *p*)) ∗ ((*c* − *q*) / (*a* − *q*))
using *daq dbr dcp* **by** (*simp add*: *field-simps*)
also have . . . ∈ ℝ
using *side1 side2 side3* **by** (*intro Reals-mult*)
finally show *?thesis* **.**
qed
— hence the third cross ratio is real
have *prod*: *complex-cross-ratio m a q r* ∗ *complex-cross-ratio m b r p* ∗ *complex-cross-ratio m c p q*
= (*a* − *r*) ∗ (*b* − *p*) ∗ (*c* − *q*) / ((*a* − *q*) ∗ (*b* − *r*) ∗ (*c* − *p*))
using *miquel-complex-cross-ratio-product*[*OF dmr daq dmp dbr dmq dcp*] **.**
have *cr1ne*: *complex-cross-ratio m a q r* ≠ *0*
unfolding *complex-cross-ratio-def* **using** *dmq dar* **by** (*simp add*: *dmr daq*)
have *cr2ne*: *complex-cross-ratio m b r p* ≠ *0*
unfolding *complex-cross-ratio-def* **using** *dmr dbp* **by** (*simp add*: *dmp dbr*)
have *cr3real*: *complex-cross-ratio m c p q* ∈ ℝ
proof −
have *complex-cross-ratio m c p q*
= ((*a* − *r*) ∗ (*b* − *p*) ∗ (*c* − *q*) / ((*a* − *q*) ∗ (*b* − *r*) ∗ (*c* − *p*)))
/ (*complex-cross-ratio m a q r* ∗ *complex-cross-ratio m b r p*)
proof −
have *ne*: *complex-cross-ratio m a q r* ∗ *complex-cross-ratio m b r p* ≠ *0*
using *cr1ne cr2ne* **by** *simp*
have *complex-cross-ratio m c p q*
= (*complex-cross-ratio m a q r* ∗ *complex-cross-ratio m b r p* ∗ *complex-cross-ratio m c p q*)

```
          / (complex-cross-ratio m a q r * complex-cross-ratio m b r p)
        using ne by simp
      then show ?thesis using prod by simp
    qed
    also have ... ∈ ℝ
      using Phi Reals-mult[OF cr1 cr2] by (rule Reals-divide)
    finally show ?thesis .
  qed
  — the converse criterion places m on the third circle
  have oncircle o3 R3 m
    using complex-cross-ratio-real-on-circle[OF c3(2) c3(3) c3(4) c3(1) dcp dcq
dpq dmq cr3real] .
  with c3 show ?thesis
    by (auto simp: concyclic-def)
qed
```

9 Concurrency of the three circumcircles

We now prove Miquel's theorem in its classical concurrency form: for a nondegenerate triangle the three circumcircles have a common point. The circumcircles of aqr and brp both pass through r. When they are not tangent there, they meet again in a second point, the *Miquel point*, which is the reflection of r in the line joining the two centres. Reflection in a line is an isometry fixing every point of that line, so this second point automatically lies on both circles; the pivot theorem then places it on the third.

```
definition reflect-line :: complex ⇒ complex ⇒ complex ⇒ complex
  where reflect-line o1 o2 z = o1 + (o2 − o1) * cnj (z − o1) / cnj (o2 − o1)
```

Reflection in the line through *o1* and *o2* preserves the distance to *o1*.

```
lemma reflect-line-dist1:
  assumes oo: o1 ≠ o2
  shows cmod (reflect-line o1 o2 z − o1) = cmod (z − o1)
proof −
  have cmod (reflect-line o1 o2 z − o1)
      = cmod ((o2 − o1) * cnj (z − o1) / cnj (o2 − o1))
    by (simp add: reflect-line-def)
  also have ... = cmod (o2 − o1) * cmod (z − o1) / cmod (o2 − o1)
    by (simp add: norm-mult norm-divide complex-mod-cnj del: complex-cnj-diff)
  also have ... = cmod (z − o1)
  proof −
    have cmod (o2 − o1) ≠ 0 using oo by (metis norm-eq-zero right-minus-eq)
    thus ?thesis by simp
  qed
  finally show ?thesis .
qed
```

Reflection in the line through *o1* and *o2* also preserves the distance to *o2*.

```
lemma reflect-line-dist2:
  assumes oo: o1 ≠ o2
  shows cmod (reflect-line o1 o2 z − o2) = cmod (z − o2)
proof −
  have g0: cnj (o2 − o1) ≠ 0
    using oo by (metis complex-cnj-zero-iff right-minus-eq)
  have cnjrel: cnj (z − o1) = cnj (z − o2) + cnj (o2 − o1)
  proof −
    have cnj (z − o1) = cnj ((z − o2) + (o2 − o1)) by simp
   also have . . . = cnj (z − o2) + cnj (o2 − o1) by (simp add: complex-cnj-add)
    finally show ?thesis .
  qed
  have eqform: reflect-line o1 o2 z − o2 = (o2 − o1) ∗ cnj (z − o2) / cnj (o2 −
o1)
  proof −
    have (reflect-line o1 o2 z − o2) ∗ cnj (o2 − o1)
        = (o1 − o2) ∗ cnj (o2 − o1) + (o2 − o1) ∗ cnj (z − o1)
      using g0 by (simp add: reflect-line-def field-simps)
    also have . . . = (o1 − o2) ∗ cnj (o2 − o1)
                  + (o2 − o1) ∗ (cnj (z − o2) + cnj (o2 − o1))
      by (simp add: cnjrel)
    also have . . . = (o2 − o1) ∗ cnj (z − o2) by (simp add: algebra-simps)
    finally have (reflect-line o1 o2 z − o2) ∗ cnj (o2 − o1) = (o2 − o1) ∗ cnj (z
− o2) .
    thus ?thesis using g0 by (simp add: field-simps)
  qed
  have cmod (reflect-line o1 o2 z − o2)
      = cmod ((o2 − o1) ∗ cnj (z − o2) / cnj (o2 − o1))
    by (simp add: eqform)
  also have . . . = cmod (o2 − o1) ∗ cmod (z − o2) / cmod (o2 − o1)
    by (simp add: norm-mult norm-divide complex-mod-cnj del: complex-cnj-diff)
  also have . . . = cmod (z − o2)
  proof −
    have cmod (o2 − o1) ≠ 0 using oo by (metis norm-eq-zero right-minus-eq)
    thus ?thesis by simp
  qed
  finally show ?thesis .
qed
```

If r does not lie on the line through the two centres, i.e. the two circles are not tangent at r, then the reflected point is genuinely distinct from r.

```
lemma reflect-line-neq:
  assumes ncl: ¬ collinear o1 o2 r
  shows reflect-line o1 o2 r ≠ r
proof
  assume eq: reflect-line o1 o2 r = r
  have oo: o1 ≠ o2
  proof
    assume o1 = o2
```

```
    hence collinear o1 o2 r by (simp add: collinear-iff-cross)
    with ncl show False by simp
  qed
  have g0: cnj (o2 − o1) ≠ 0
    using oo by (metis complex-cnj-zero-iff right-minus-eq)
  have (reflect-line o1 o2 r − r) ∗ cnj (o2 − o1)
      = (o2 − o1) ∗ cnj (r − o1) − (r − o1) ∗ cnj (o2 − o1)
    using g0 by (simp add: reflect-line-def field-simps)
  moreover have reflect-line o1 o2 r − r = 0 using eq by simp
  ultimately have z0: (o2 − o1) ∗ cnj (r − o1) − (r − o1) ∗ cnj (o2 − o1) =
0
    by simp
  from z0 have zeq: (o2 − o1) ∗ cnj (r − o1) = (r − o1) ∗ cnj (o2 − o1) by
simp
  have cnj (cnj (o2 − o1) ∗ (r − o1)) = (o2 − o1) ∗ cnj (r − o1)
    by (simp add: complex-cnj-mult)
  also have ... = (r − o1) ∗ cnj (o2 − o1) by (rule zeq)
  also have ... = cnj (o2 − o1) ∗ (r − o1) by (simp add: mult.commute)
  finally have cnj (cnj (o2 − o1) ∗ (r − o1)) = cnj (o2 − o1) ∗ (r − o1) .
  hence Im (cnj (o2 − o1) ∗ (r − o1)) = 0 by (metis cnj.sel(2) neg-equal-zero)
  hence collinear o1 o2 r by (simp add: collinear-iff-cross)
  with ncl show False by simp
qed
```

The classical Miquel theorem. Let abc be a nondegenerate triangle with r, p, q on the side lines ab, bc, ca. Given the circumcircles of aqr and brp, not tangent at r, their second common point — the reflection of r in the centre line — lies on all three circumcircles, including that of cpq. No common point is assumed: it is constructed.

```
theorem miquel-concurrent:
  assumes lr: on-line a b r and lp: on-line b c p and lq: on-line c a q
    and ncol: ¬ collinear a b c
    and ca: oncircle o1 R1 a and cq: oncircle o1 R1 q
    and car: oncircle o1 R1 r and R1pos: 0 < R1
    and cbb: oncircle o2 R2 b and cbr: oncircle o2 R2 r
    and cpp: oncircle o2 R2 p and R2pos: 0 < R2
    and nonline: ¬ collinear o1 o2 r
    and daq: a ≠ q and dar: a ≠ r and dbr: b ≠ r and dbp: b ≠ p
    and dcp: c ≠ p and dcq: c ≠ q
  shows concyclic a q r (reflect-line o1 o2 r)
      ∧ concyclic b r p (reflect-line o1 o2 r)
      ∧ concyclic c p q (reflect-line o1 o2 r)
proof −
  define m where m = reflect-line o1 o2 r
  have oo: o1 ≠ o2
  proof
    assume o1 = o2
    hence collinear o1 o2 r by (simp add: collinear-iff-cross)
    with nonline show False by simp
```

```
qed
— the constructed point lies on the first two circles
have m1: oncircle o1 R1 m
proof −
  have cmod (m − o1) = cmod (r − o1)
    unfolding m-def by (rule reflect-line-dist1[OF oo])
  also have ... = R1 using car by (simp add: oncircle-def)
  finally show ?thesis by (simp add: oncircle-def)
qed
have m2: oncircle o2 R2 m
proof −
  have cmod (m − o2) = cmod (r − o2)
    unfolding m-def by (rule reflect-line-dist2[OF oo])
  also have ... = R2 using cbr by (simp add: oncircle-def)
  finally show ?thesis by (simp add: oncircle-def)
qed
have h1: concyclic a q r m
  unfolding concyclic-def using R1pos ca cq car m1 by blast
have h2: concyclic b r p m
  unfolding concyclic-def using R2pos cbb cbr cpp m2 by blast
have mr: m ≠ r unfolding m-def by (rule reflect-line-neq[OF nonline])
— the third circle exists because the triangle is nondegenerate
have ncolcpq: ¬ collinear c p q
  using cpq-ncol[OF ncol lp lq] dcp dcq by auto
have c3: concyclic3 c p q using circumcircle-exists[OF ncolcpq] .
— the constructed point lies on the third circle
have h3: concyclic c p q m
proof (cases m = p)
  case True
  from c3 obtain ctr R where 0 < R
      oncircle ctr R c oncircle ctr R p oncircle ctr R q
    by (auto simp: concyclic3-def)
  thus ?thesis using True unfolding concyclic-def by blast
next
  case notp: False
  show ?thesis
  proof (cases m = q)
    case True
    from c3 obtain ctr R where 0 < R
        oncircle ctr R c oncircle ctr R p oncircle ctr R q
      by (auto simp: concyclic3-def)
    thus ?thesis using True unfolding concyclic-def by blast
  next
    case notq: False
    show ?thesis
      using miquel-pivot[OF lr lp lq h1 h2 ncol notp notq mr
                      daq dar dbr dbp dcp dcq] .
  qed
qed
```

show *?thesis* **using** *h1 h2 h3* **by** (*simp add*: *m-def*)
qed

The existence form: the three circumcircles have a common point.

corollary *miquel-point-exists*:
assumes *lr*: *on-line a b r* **and** *lp*: *on-line b c p* **and** *lq*: *on-line c a q*
and *ncol*: ¬ *collinear a b c*
and *ca*: *oncircle o1 R1 a* **and** *cq*: *oncircle o1 R1 q*
and *car*: *oncircle o1 R1 r* **and** *R1pos*: *0* < *R1*
and *cbb*: *oncircle o2 R2 b* **and** *cbr*: *oncircle o2 R2 r*
and *cpp*: *oncircle o2 R2 p* **and** *R2pos*: *0* < *R2*
and *nonline*: ¬ *collinear o1 o2 r*
and *daq*: *a* ≠ *q* **and** *dar*: *a* ≠ *r* **and** *dbr*: *b* ≠ *r* **and** *dbp*: *b* ≠ *p*
and *dcp*: *c* ≠ *p* **and** *dcq*: *c* ≠ *q*
shows ∃ *m*. *concyclic a q r m* ∧ *concyclic b r p m* ∧ *concyclic c p q m*
using *miquel-concurrent*[*OF lr lp lq ncol ca cq car R1pos cbb cbr cpp R2pos nonline daq dar dbr dbp dcp dcq*]
by *blast*

end

References

[1] H. S. M. Coxeter and S. L. Greitzer. *Geometry Revisited*, volume 19 of *New Mathematical Library*. Mathematical Association of America, 1967.

[2] R. A. Johnson. *Modern Geometry: An Elementary Treatise on the Geometry of the Triangle and the Circle*. Houghton Mifflin, Boston, 1929.

[3] F. Marić and D. Simić. Complex geometry. *Archive of Formal Proofs*, 2019. https://isa-afp.org/entries/Complex_Geometry.html, Formal proof development.

[4] A. Miquel. Théorèmes de géométrie. *Journal de mathématiques pures et appliquées*, 3:485–487, 1838. Available online at https://gallica.bnf.fr/ark:/12148/bpt6k16382m.f493.

[5] A. F. Ramos, D. B. Hulak, and R. J. G. B. de Queiroz. The wallace–simson line theorem in isabelle/hol. *Archive of Formal Proofs*, August 2026. https://isa-afp.org/entries/Simson.html, Formal proof development.